# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=cln
pkgver=1.3.6
pkgrel=2
pkgdesc="Class library for numbers"
arch=('x86_64')
url="https://www.ginac.de/CLN/"
license=('GPL')
depends=('gmp')
makedepends=('texlive-core')
source=(https://www.ginac.de/CLN/${pkgname}-${pkgver}.tar.bz2)
sha1sums=('144f15a57f4b25ada0f10ff28458de03578f6f16')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make all html pdf
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install install-html install-pdf
}
